
1. [bookmark: _Toc122434485][bookmark: _Toc126242285]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk37 related to the title of this CR.
Contact:	Shaun Harry 
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc126242286]Corrections required

1. [bookmark: _Toc54888063][bookmark: _Toc100131355][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	altstep a_NR_RacingCond_AwaitRrcMessageAndTwoSysInd

	Reason for change
	Check for received SYSTEM_IND1 is using variable p_SysIndByRef2 

	Summary of change
	Correct check for “RRC message and/or SYSTEM_IND1 has not been received yet”

	TTCN module
	NR_Timing.ttcn

	MCC160 Comment
	


[bookmark: _Toc54888064]
Before Change:
	
altstep a_NR_RacingCond_AwaitRrcMessageAndTwoSysInd(NR_CellId_Type p_NR_CellId,
                                                      RacingCond_ExpectedSequence_Type p_ExpectedSequence,
                                                      inout template (omit) NR_SRB_COMMON_IND p_SrbIndByRef,
                                                      inout template (omit) NR_SYSTEM_IND p_SysIndByRef1,
                                                      inout template (omit) NR_SYSTEM_IND p_SysIndByRef2,
                                                      template (present) NR_SRB_COMMON_IND p_ExpectedSrbInd := car_NR_SRB_IND_Any,
                                                      template (present) NR_SYSTEM_IND p_ExpectedSysInd1 := car_NR_SYSTEM_IND_Any,
                                                      template (present) NR_SYSTEM_IND p_ExpectedSysInd2 := car_NR_SYSTEM_IND_Any)
                                                   runs on NR_BASE_PTC
  { /* note: since this is an altstep 'interleave' cannot be used */
    /* note: since the altstep uses repeat to receive the second message the altstep is left and re-entered again
     *       => local variables are uninitialised; the only way to store information of subsequent messages is by the inout parameters
     *          (which for that reason have to be "inout"; "out" would not work */
    var NR_SRB_COMMON_IND v_SRB_COMMON_IND;
    var NR_SYSTEM_IND v_SYSTEM_IND;
    var SubFrameTiming_Type v_Timing_SysInd1;
    var SubFrameTiming_Type v_Timing_SysInd2;
    var SubFrameTiming_Type v_Timing_SrbInd;

    [not isvalue(p_SrbIndByRef)] SRB.receive (p_ExpectedSrbInd) -> value v_SRB_COMMON_IND
      {
        p_SrbIndByRef := v_SRB_COMMON_IND;
        if (not (isvalue(p_SysIndByRef1) and isvalue(p_SysIndByRef2))) {// both SYSTEM_INDs not been received yet
          repeat;                                                       // => wait for SYSTEM_IND1 and/or 2
        } else {
          v_Timing_SrbInd := valueof(p_SrbIndByRef.Common.TimingInfo.SubFrame);
          v_Timing_SysInd1 := valueof(p_SysIndByRef1.Common.TimingInfo.SubFrame);
          v_Timing_SysInd2 := valueof(p_SysIndByRef2.Common.TimingInfo.SubFrame);
          fl_NR_RacingCond_CheckSequence(p_NR_CellId, p_ExpectedSequence, v_Timing_SrbInd, v_Timing_SysInd1, v_Timing_SysInd2);
        }
      }
    [not isvalue(p_SysIndByRef1)] SYSIND.receive (p_ExpectedSysInd1) -> value v_SYSTEM_IND
      {
        p_SysIndByRef1 := v_SYSTEM_IND;
        if (not (isvalue(p_SrbIndByRef) and isvalue(p_SysIndByRef2))) { // RRC message and/or SYSTEM_IND2 has not been received yet
          repeat;                                                       // => wait for RRC PDU and/or SYSTEM_IND2
        } else {
          v_Timing_SrbInd := valueof(p_SrbIndByRef.Common.TimingInfo.SubFrame);
          v_Timing_SysInd1 := valueof(p_SysIndByRef1.Common.TimingInfo.SubFrame);
          v_Timing_SysInd2 := valueof(p_SysIndByRef2.Common.TimingInfo.SubFrame);
          fl_NR_RacingCond_CheckSequence(p_NR_CellId, p_ExpectedSequence, v_Timing_SrbInd, v_Timing_SysInd1, v_Timing_SysInd2);
        }
      }
    [not isvalue(p_SysIndByRef2)] SYSIND.receive (p_ExpectedSysInd2) -> value v_SYSTEM_IND
      {
        p_SysIndByRef2 := v_SYSTEM_IND;
        if (not (isvalue(p_SrbIndByRef) and isvalue(p_SysIndByRef2))) { // RRC message and/or SYSTEM_IND1 has not been received yet
          repeat;                                                       // => wait for RRC PDU
        } else {
          v_Timing_SrbInd := valueof(p_SrbIndByRef.Common.TimingInfo.SubFrame);
          v_Timing_SysInd1 := valueof(p_SysIndByRef1.Common.TimingInfo.SubFrame);
          v_Timing_SysInd2 := valueof(p_SysIndByRef2.Common.TimingInfo.SubFrame);
          fl_NR_RacingCond_CheckSequence(p_NR_CellId, p_ExpectedSequence, v_Timing_SrbInd, v_Timing_SysInd1, v_Timing_SysInd2);
        }
      }
  }
  



After Change:
	
altstep a_NR_RacingCond_AwaitRrcMessageAndTwoSysInd(NR_CellId_Type p_NR_CellId,
                                                      RacingCond_ExpectedSequence_Type p_ExpectedSequence,
                                                      inout template (omit) NR_SRB_COMMON_IND p_SrbIndByRef,
                                                      inout template (omit) NR_SYSTEM_IND p_SysIndByRef1,
                                                      inout template (omit) NR_SYSTEM_IND p_SysIndByRef2,
                                                      template (present) NR_SRB_COMMON_IND p_ExpectedSrbInd := car_NR_SRB_IND_Any,
                                                      template (present) NR_SYSTEM_IND p_ExpectedSysInd1 := car_NR_SYSTEM_IND_Any,
                                                      template (present) NR_SYSTEM_IND p_ExpectedSysInd2 := car_NR_SYSTEM_IND_Any)
                                                   runs on NR_BASE_PTC
  { /* note: since this is an altstep 'interleave' cannot be used */
    /* note: since the altstep uses repeat to receive the second message the altstep is left and re-entered again
     *       => local variables are uninitialised; the only way to store information of subsequent messages is by the inout parameters
     *          (which for that reason have to be "inout"; "out" would not work */
    var NR_SRB_COMMON_IND v_SRB_COMMON_IND;
    var NR_SYSTEM_IND v_SYSTEM_IND;
    var SubFrameTiming_Type v_Timing_SysInd1;
    var SubFrameTiming_Type v_Timing_SysInd2;
    var SubFrameTiming_Type v_Timing_SrbInd;

    [not isvalue(p_SrbIndByRef)] SRB.receive (p_ExpectedSrbInd) -> value v_SRB_COMMON_IND
      {
        p_SrbIndByRef := v_SRB_COMMON_IND;
        if (not (isvalue(p_SysIndByRef1) and isvalue(p_SysIndByRef2))) {// both SYSTEM_INDs not been received yet
          repeat;                                                       // => wait for SYSTEM_IND1 and/or 2
        } else {
          v_Timing_SrbInd := valueof(p_SrbIndByRef.Common.TimingInfo.SubFrame);
          v_Timing_SysInd1 := valueof(p_SysIndByRef1.Common.TimingInfo.SubFrame);
          v_Timing_SysInd2 := valueof(p_SysIndByRef2.Common.TimingInfo.SubFrame);
          fl_NR_RacingCond_CheckSequence(p_NR_CellId, p_ExpectedSequence, v_Timing_SrbInd, v_Timing_SysInd1, v_Timing_SysInd2);
        }
      }
    [not isvalue(p_SysIndByRef1)] SYSIND.receive (p_ExpectedSysInd1) -> value v_SYSTEM_IND
      {
        p_SysIndByRef1 := v_SYSTEM_IND;
        if (not (isvalue(p_SrbIndByRef) and isvalue(p_SysIndByRef2))) { // RRC message and/or SYSTEM_IND2 has not been received yet
          repeat;                                                       // => wait for RRC PDU and/or SYSTEM_IND2
        } else {
          v_Timing_SrbInd := valueof(p_SrbIndByRef.Common.TimingInfo.SubFrame);
          v_Timing_SysInd1 := valueof(p_SysIndByRef1.Common.TimingInfo.SubFrame);
          v_Timing_SysInd2 := valueof(p_SysIndByRef2.Common.TimingInfo.SubFrame);
          fl_NR_RacingCond_CheckSequence(p_NR_CellId, p_ExpectedSequence, v_Timing_SrbInd, v_Timing_SysInd1, v_Timing_SysInd2);
        }
      }
    [not isvalue(p_SysIndByRef2)] SYSIND.receive (p_ExpectedSysInd2) -> value v_SYSTEM_IND
      {
        p_SysIndByRef2 := v_SYSTEM_IND;
        if (not (isvalue(p_SrbIndByRef) and isvalue(p_SysIndByRef1))) { // RRC message and/or SYSTEM_IND1 has not been received yet
          repeat;                                                       // => wait for RRC PDU
        } else {
          v_Timing_SrbInd := valueof(p_SrbIndByRef.Common.TimingInfo.SubFrame);
          v_Timing_SysInd1 := valueof(p_SysIndByRef1.Common.TimingInfo.SubFrame);
          v_Timing_SysInd2 := valueof(p_SysIndByRef2.Common.TimingInfo.SubFrame);
          fl_NR_RacingCond_CheckSequence(p_NR_CellId, p_ExpectedSequence, v_Timing_SrbInd, v_Timing_SysInd1, v_Timing_SysInd2);
        }
      }
  }
  




